
C *********************************************************************
C *                                                                   *
C *                           J E M F H X                             *
C *                                                                   *
C *              Jet Emission/Faraday Rotn for Helical Fields         *
C *                 With Lobe Background and Surface Shear            *
C *                                                                   *
C *                         Sun Version 1.1                           *
C *                     A.H.Bridle    04-Oct-95                       *
C *                                                                   *
C * finds I, Q AND U profiles across helical-field cylindrical jet    *
C * whose axis is along the z-axis.  Profiles calculated on NL lines  *
C * of sight by cell summation for Stokes I, and by cell-by-cell      *
C * rotation of Q and U for Rotation Measure                          *
C *                                                                   *
C * can have arbitrary inclination I of jet to plane of sky           *
C * can superpose jet on a polarized background lobe (constant)       *
C *                                                                   *
C * coded for nonrelativistic bulk motion                             *
C *           constant electron energy spectrum through jet           *
C *                                                                   *
C *           ADJUSTABLE FIELD CONFIGURATION (SUBROUTINE B3D)         *
C *           CAN BE CHAN/HENRIKSEN, SIMPLE HELIX OR SHEARED HELIX    *
C *           WITH RANDOM FIELD COMPONENT ADDED AS REQUIRED           *
C *           PLUS AN OUTER SHEAR LAYER WITH PURE BZ                  *
C *                                                                   *
C * device assignments are:                                           *
C *                           3: print file ---> fort.3               *
C *                           5: screen input                         *
C *                           6: screen output                        *
C *                           8: disk file output (you name it)       *
C *********************************************************************
C
      REAL INORM,IMAX,IBGRD,JETI,JETID
      CHARACTER*24 DAT
      CHARACTER*12 DISKNAME
      COMMON /AREA1/GAMMA,R,RHDIST,IHDIST,IPITCH,ILIST
      COMMON /AREA2/SINI,COSI,PSIR,SPSIR,CPSIR,ROT,BRAND,IFIELD
      COMMON /AREA3/PROF(5,136),PINT
      COMMON /AREA4/PI,ICDIST,RCDIST
      COMMON /AREA5/BSHEAR,RSHEAR,IBGRD,PBGRD,CHIBGRD
      LOGICAL DEBUG
      COMMON /DEBUG/DEBUG
      DEBUG=.FALSE.
C
C  PROFILE ARRAY IS DEFINED TO 136 TO ALLOW PARM STORAGE ON DISK
C
      PI=3.14159265
      CALL FDATE(DAT)
      WRITE(6,5) DAT
    5 FORMAT(' A.H.B. JET EMISSION MODEL WITH FARADAY EFFECTS',/,/,
     *' coded 04-Oct-95, EXECUTION on ',A24,/,
     *' ******************************************************',/)
      IF(DEBUG.EQ.(.TRUE.)) WRITE(6,770)
  770 FORMAT(' !!! DEBUGGING TURNED ON - CONTROL-C OUT WHEN ENOUGH !!!')
C
C     ZERO THE FLUX-DENSITY ARRAYS
C
   10 DO 12 J=1,5



      DO 11 K=1,136
      PROF(J,K)=0.
   11 CONTINUE
   12 CONTINUE
C======================================================================
C
C          SPECIFY THE CONSTANTS OF JET AND ITS FIELD
C
C          PSIR IS REFERENCE PITCH OF FIELD
C                            CH  - ARCTAN(BZ/PHI) AT SURFACE
C                            HELIX - CONSTANT VALUE OF ARCTAN(BZ/BPHI)
C                            SH. HELIX - ARCTAN(BZ/BPHI) AT CENTER
C               I.E. ANGLE BETWEEN FIELD LINE AND JET AZIMUTH CIRCLE
C          PSIRD - DEGREES    PSIR - RADIANS
C
C          BRAND IS MEAN AMPLITUDE OF RANDOM FIELD RELATIVE TO MAXIMUM
C              VALUE OF ORGANISED FIELD
C
C          GAMMA IS HOT ELECTRON ENERGY INDEX
C          R IS JET BDRY RADIUS (ARBITRARY LENGTH SCALE)
C          NL IS NUMBER OF LINES OF SIGHT
C          JETI IS INCLINATION OF JET TO SKY IN RADIANS
C
C          RSHEAR  IS INNER RADIUS OF SHEAR LAYER
C          BSHEAR  IS FIELD STRENGTH IN SHEAR LAYER
C          IBGRD   IS INTENSITY IN BACKGROUND
C          PBGRD   IS DEGREE OF POLARIZATION IN BACKGROUND
C          CHIBGRD IS POLARIZATION POSITION ANGLE IN BACKGROUND
C
C
C======================================================================
C
      WRITE(6,13)
   13 FORMAT(' BEGIN INPUTS:',/,
     *' INCLINATION OF JET TO SKY IN DEG (F6.2, <90) ?')
      READ(5,16) JETID
      IF (ABS(JETID).EQ.90.0) GO TO 777
C
      WRITE(6,14)
   14 FORMAT(' REFERENCE PITCH OF FIELD IN DEG (F6.2) ?')
      READ(5,16) PSIRD
C
C      WRITE(6,15)
   15 FORMAT(' GAMMA(F6.2) ?')
C      READ(5,16) GAMMA
      GAMMA=2.3
   16 FORMAT(F6.2)
C
      WRITE(6,555)
  555 FORMAT(' MEAN RANDOM FIELD (MAX ORGANISED=1) ?')
      READ(5,16) BRAND
C
      WRITE(6,556)
  556 FORMAT(' TYPE OF FIELD: 1=CH  2=HELIX  3=SHEARED HELIX')
      READ(5,20) IFIELD
C
C======================================================================
C
C     SPECIFY TWO TYPES OF ELECTRON DISTRIBUTION IN JET



C
C======================================================================
C
C     HOT ELECTRONS - COMPONENT CONTRIBUTING SYNCHROTRON EMISSION
C     IHDIST SPECIFIES TYPE OF DISTRIBUTION
C     RHDIST SPECIFIES AN APPROPRIATE RADIAL SCALE FACTOR
C     IPITCH SPECIFIES PITCH ANGLE DISTRIBUTION RELATIVE TO MAGN FIELD
C
C======================================================================
C
C      WRITE(6,17)
   17 FORMAT(' HOT ELECTRON DISTRIB: 1=GAUSS 2=EXP 3=TUBE ?')
C      READ(5,20) IHDIST
       IHDIST=3
C
C      WRITE(6,18)
   18 FORMAT(' SCALE(F6.2): 1/ESQ RADIUS (GAUSS), SCALE RADIUS (EXP)'
     *,' INNER RADIUS (TUBE) ?')
C      READ(5,16) RHDIST
       RHDIST=0
C
C     SELECT PITCH ANGLE DISTRIBUTION FUNCTION
C
C      WRITE(6,19)
   19 FORMAT(' PITCH ANGLE DISTRIB: 1=UNIFORM 2=COS 3=COSSQ ?')
C      READ(5,20) IPITCH
      IPITCH=1
C======================================================================
C
C     COLD ELECTRONS - COMPONENT CONTRIBUTING FARADAY ROTATION
C
C     ROT SPECIFIES THE FARADAY DEPTH OF A CELL WITH NMAX,BMAX
C     ICDIST SPECIFIES TYPE OF DISTRIBUTION
C     RCDIST SPECIFIES AN APPROPRIATE RADIAL SCALE FACTOR
C
C======================================================================
C      WRITE(6,598)
  598 FORMAT(' ROTATION MEASURE PARM (F6.2) ?')
C      READ(5,16) ROT
      ROT=0.
C
C      WRITE(6,509)
  509 FORMAT(' COLD ELECTRON DISTRIB: 1=GAUSS 2=EXP 3=TUBE ?')
C      READ(5,20) ICDIST
      ICDIST=3
C
C      WRITE(6,508)
  508 FORMAT(' SCALE(F6.2): 1/ESQ RADIUS (GAUSS), SCALE RADIUS (EXP)'
     *,' INNER RADIUS (TUBE) ?')
C      READ(5,16) RCDIST
       RCDIST=0.
C
C=====================================================================
C
C     SET UP THE SHEAR LAYER PARAMETERS
C
C=====================================================================
C
      WRITE(6,510) 



  510 FORMAT(' SHEAR LAYER FIELD STRENGTH (F6.2)?')
      READ(5,16)BSHEAR
      WRITE(6,511)
  511 FORMAT(' SHEAR LAYER INNER RADIUS (JET=60) (F6.2)?')
      READ(5,16)RSHEAR
C
C
C====================================================================
C
C     SET UP THE BACKGROUND PARAMETERS
C
C====================================================================
C
C
      WRITE(6,520)
 520  FORMAT(' BACKGROUND INTENSITY (F6.2)?')
      READ(5,16)IBGRD
      WRITE(6,521)
 521  FORMAT(' BACKGROUND FRACTIONAL POLARIZATION (F6.2)?')
      READ(5,16)PBGRD
      WRITE(6,522)
 522  FORMAT(' BACKGROUND POLARIZATION POSITION ANGLE (F6.2)?')
      READ(5,16)CHIBGRD
C
C===================================================================
C
C     SET UP RADIUS AND SIGHTLINE PARAMETERS
C
C===================================================================
C
      R=60.
      NL=131
C
C===================================================================
C
C     SELECT OUTPUT OPTIONS FOR LPT
C
C===================================================================
C
      WRITE(6,21)
   21 FORMAT(' LINE FLUX LIST ON LPT ?  (0=NO  1=YES)')
      READ(5,20) ILIST
   20 FORMAT(I1)
C
C===================================================================
C
C     MODEL COMPUTATIONS BEGIN HERE 
C
C===================================================================
C
C
 999  CONTINUE
CRC     INTRINSIC PERCENTAGE POLARIZATION (CONST. THROUGH JET HERE)
C
      PINT=(GAMMA+1.)/(GAMMA+7./3.)
C
C     INPUT ANGLES DEGREES => RADIANS AND COMPUTE SINE, COSINE
C
      PSIR=PSIRD*PI/180.



      SPSIR=SIN(PSIR)
      CPSIR=COS(PSIR)
      JETI=JETID*PI/180.
      SINI=SIN(JETI)
      COSI=COS(JETI)
      IF(ILIST.NE.2) GO TO 998
      WRITE(6,911) SPSIR,CPSIR
 911  FORMAT(2E10.3)
C
C     COMPUTE CELL WIDTH = W
C
  998 W=4.*R/(2.*NL-1.)
      IF (ILIST.EQ.0) GO TO 24
C
C     Print the model parameters and the execution date
C
      CALL PRINT(JETID,PSIRD,GAMMA,R,IHDIST,RHDIST,IPITCH,
     *ROT,BRAND,IFIELD,ICDIST,RCDIST,DAT,RSHEAR,BSHEAR,
     *IBGRD,PBGRD,CHIBGRD,IMAX)
C
C     COMPUTE THE LINE-OF-SIGHT INTENSITIES AND FIND THEIR MAXIMUM
C
   24 IMAX=0.
      DO 28 N=1,NL
      DN=(N-66.)*W
      PROF(1,N)=DN
      CALL LINE(DN,W,NL,PROF(2,N),PROF(3,N),PROF(4,N),PROF(5,N))
C
C     CHECK IF PROF(2,N) IS MAXIMUM TOTAL INTENSITY SO FAR
C     RESET IMAX IF IT IS
C
      IF(PROF(2,N).GT.IMAX) IMAX=PROF(2,N)
   28 CONTINUE
      IF (ILIST.EQ.0) GO TO 30
      DO 29 N=1,NL
C======================================================================
C
C     PRINT THE NORMALISED OUTPUT FROM THIS LINE OF SIGHT
C
C======================================================================
      INORM=PROF(2,N)/IMAX
      QNORM=PROF(3,N)/PROF(2,N)
      UNORM=PROF(4,N)/PROF(2,N)
      PNORM=SQRT(QNORM*QNORM+UNORM*UNORM)
      PA=ATAN2(UNORM,QNORM)*180./(2.*PI)
      FDEPT=PROF(5,N)
      WRITE(3,26) N,PROF(1,N),INORM,PNORM,PA,FDEPT
   26 FORMAT('
 N=',I3,'     DN=',F10.3,5X,'I=',F10.5,
     * 5X,'P=',F7.2,'%, PA=',F6.2,' TO JET',5X,'FARADAY DEPTH=',F6.2)
   29 CONTINUE
      WRITE(3,55)
      WRITE(3,55)
      WRITE(3,55)
   55 FORMAT(' ')
C======================================================================
C 
C      PLOT THE NORMALISED OUTPUT FOR THIS LINE OF SIGHT
C



C======================================================================
C
CALL THE ROUTINE FOR LPT PLOT OF INTENSITY AND %POL PROFILES
C
   30 CALL LPLOT(NL,IMAX)
C
C     PRINT MODEL PARAMETERS BELOW PLOT
C
      CALL PRINT(JETID,PSIRD,GAMMA,R,IHDIST,RHDIST,IPITCH,
     *ROT,BRAND,IFIELD,ICDIST,RCDIST,DAT,RSHEAR,BSHEAR,
     *IBGRD,PBGRD,CHIBGRD,IMAX)
C
C
C     WRITE INTENSITY AND POLARIZATION PROFILES TO DISK ?
C
      WRITE(6,503)
  503 FORMAT(' WRITE PROFILES TO DISK ?  0=NO 1=YES ?')
      READ(5,20) IDISK
      IF (IDISK.EQ.0) GO TO 99
C
C     LOAD HIGH END OF THE EMISSION PROFILE ARRAY WITH MODEL PARMS
C     SO WE CAN IDENTIFY IT LATER
C
      PROF(1,132)=R
      PROF(2,132)=IMAX
      PROF(3,132)=GAMMA
      PROF(4,132)=PSIRD
      PROF(1,133)=JETID
      PROF(2,133)=ROT
      PROF(3,133)=RHDIST
      PROF(4,133)=RCDIST
      PROF(1,134)=FLOAT(NL)
      PROF(2,134)=FLOAT(IHDIST)
      PROF(3,134)=FLOAT(IPITCH)
      PROF(4,134)=FLOAT(ICDIST)
      PROF(1,135)=BRAND
      PROF(2,135)=FLOAT(IFIELD)
      PROF(3,135)=BSHEAR
      PROF(4,135)=RSHEAR
      PROF(1,136)=IBGRD
      PROF(2,136)=PBGRD
      PROF(3,136)=CHIBGRD
C
C    DISK OPENING ROUTINE 
C
      WRITE(6,31)
   31 FORMAT(' NAME OF DISK OUTFILE ?')
      READ(5,32) DISKNAME
   32 FORMAT(A12)
      OPEN (UNIT=8, FILE=DISKNAME)
C======================================================================
C
C     WRITE THE ARRAY TO DISK AS A TABLE
C
C======================================================================
      DO 33 J=1,131
   33 WRITE(8,34) PROF(1,J),SQRT(PROF(3,J)**2+PROF(4,J)**2),
     *PROF(2,J),PROF(3,J),PROF(4,J),PROF(5,J)
   34 FORMAT(6F10.3)



C
C     CLOSE THE DISK
C
      CLOSE (UNIT=8)
C
C     CHECK WHETHER WE WANT TO STOP HERE
C
   99 CALL REVISE(PSIRD,JETID,NL,IEX)
      IF (IEX.EQ.3) GO TO 1
      GO TO 999
    1 STOP
C======================================================================
C
C     COMPLAIN ABOUT ILLEGAL INPUTS
C
C
C     IMPROPER INCLINATION = 90 DEG
C
  777 WRITE(6,778)
  778 FORMAT(' ILLEGAL INCLINATION - GIVES INFINITE JET DEPTH')
      GO TO 10
C
C=====================================================================
C
      END
C
C                        ************************
C                        *                      *
C                        *        L I N E       *
C                        *                      *
C                        ************************
C
C
      SUBROUTINE LINE(DN,W,NL,ILINE,QLINE,ULINE,FLINE)
C
C     COMPUTES SUM OF CELL INTENSITIES ALONG LINE OF SIGHT DN FROM AXIS
C              FARADAY-ROTATED Q AND U INTENSITIES
C              FARADAY DEPTH
C
      REAL L,LLIM,ILINE,ICELL,IBGRD
      COMMON /AREA1/GAMMA,R,RHDIST,IHDIST,IPITCH,ILIST
      COMMON /AREA2/SINI,COSI,PSIR,SPSIR,CPSIR,ROT,BRAND,IFIELD
      COMMON /AREA3/PROF(5,136),PINT
      COMMON /AREA4/PI,ICDIST,RCDIST
      COMMON /AREA5/BSHEAR,RSHEAR,IBGRD,PBGRD,CHIBGRD
      COMMON/DEBUG/DEBUG
C==================================================================
C
      ILINE=0.
      QLINE=0.
      ULINE=0.
      FLINE=0.
      D=DN
C
C     CAN'T HAVE CELLS OUTSIDE JET
C
      IF (ABS(D).GE.R) GO TO 36
C
C     COMPUTE CLOSEST AND FURTHEST CELL DEPTHS = LLIM



C                                      NUMBERS = NLIM
C
      LLIM=SQRT(R**2-D**2)
      NLIM=INT(LLIM/W)
C
C     CELL ITERATION STARTS HERE
C     COMPUTE AND ADD ONLY THOSE CELL CONTRIBUTIONS FROM -LLIM TO +LLIM
C
      DO 11 I=1,2*NLIM+1
      IF(DEBUG.EQ.(.TRUE.)) WRITE(6,2) I
    2 FORMAT(' CALCULATING CELL I=',I2)
      L=-W*FLOAT(NLIM+1-I)
      IF (ABS(L).GT.LLIM) GO TO 36
C
C     FIRST COMPUTE CELL EMISSIVITY
C
      CALL EFCELL(L,D,W,EMISS,CHIP,FARAD)
      IF(I.EQ.1.OR.I.EQ.2*NLIM+1) GO TO 6
      ICELL=W*EMISS
      GO TO 5
    6 ICELL=(W/2.+LLIM-NLIM*W)*EMISS
    5 CONTINUE
C
C     CALCULATE CURRENT TOTAL POLARIZED INTENSITY AND ITS POSITION ANGLE
C
      IF(ULINE.EQ.0.0.AND.QLINE.EQ.0.0) GO TO 10
      CHI=ATAN2(ULINE,QLINE)/2
      GO TO 12
   10 CHI=0.0
   12 PLINE=SQRT(ULINE*ULINE+QLINE*QLINE)
C
C     COMPUTE Q AND U
C
      QCELL=ICELL*PINT*COS(2.*CHIP)
      UCELL=ICELL*PINT*SIN(2.*CHIP)
C
C     ADD THIS CELL'S CONTRIBUTION TO THE INTENSITY SUMMATION
C                                  AND TO THE FARADAY ROTATION
C
      ILINE=ILINE+ICELL
      QLINE=PLINE*COS(2.*(CHI+FARAD))+QCELL
      ULINE=PLINE*SIN(2.*(CHI+FARAD))+UCELL
      FLINE=FLINE+FARAD
      IF(DEBUG.EQ.(.TRUE.)) WRITE(6,3) I,ILINE,QLINE,ULINE,FARAD
    3 FORMAT(' TOTALS ON LINE AT CELL I=',I2,/,
     *' ILINE=',F6.2,'   QLINE=',F6.2,'   ULINE=',F6.2,'   FARAD=',
     *F6.2)
   11 CONTINUE
C
C     ADD THE BACKGROUND EMISSION
C
      ILINE=ILINE+IBGRD
      QLINE=QLINE+PBGRD*IBGRD*COS(2.*CHIBGRD*PI/180)
      ULINE=ULINE+PBGRD*IBGRD*SIN(2.*CHIBGRD*PI/180)
   36 RETURN
      END
C
C                        ************************
C                        *                      *



C                        *      E F C E L L     *
C                        *                      *
C                        ************************
C
C     COMPUTE POLAR EMISSIVITY AND FARADAY DEPTH OF CELL AT (L,D)
C
C
      SUBROUTINE EFCELL(L,D,W,EMISS,CHIP,FARAD)
C
C     D IS HEIGHT OF CELL CENTER ABOVE JET AXIS
C     L IS DEPTH OF CELL CENTER BEHIND JET AXIS
C     RCELL IS RADIUS OF CELL FROM JET AXIS
C
      REAL L,LSQ,NH0,NC0
      COMMON /AREA1/GAMMA,R,RHDIST,IHDIST,IPITCH,ILIST
      COMMON /AREA2/SINI,COSI,PSIR,SPSIR,CPSIR,ROT,BRAND,IFIELD
      COMMON /AREA4/PI,ICDIST,RCDIST
      COMMON /AREA5/BSHEAR,RSHEAR,IBGRD,PBGRD,CHIBGRD
      COMMON/DEBUG/DEBUG
      IF(DEBUG.EQ.(.TRUE.)) WRITE(6,5) L,D,W
    5 FORMAT(' CALCULATING EMISSIVITY IN CELL L=',F6.2,'   D=',F6.2,
     *'   W=',F6.2)
    
C
C     COMPUTE RADIUS OF THIS CELL IN THE JET => RCELL
C
      LSQ=L**2
      RSQ=LSQ+D**2
      RCELL=SQRT(RSQ)
C
C     CALCULATE THE B FIELD COMPONENTS IN THIS CELL
C
      CALL B3D(L,D,RCELL,BTOT,BPERP,BPARL,BX,BY,BZ)
      IF (BPERP.EQ.0.0.AND.BTOT.EQ.0.0) GO TO 10
      IF (RCELL.EQ.0.0.AND.PSIR.EQ.0.0) GO TO 10
      THETA = ASIN(BPERP/BTOT)
      GO TO 20
   10 THETA = PI/2.
   20 EMISS=(BPERP)**((GAMMA+1.)/2.)*G(THETA,IPITCH)*
     *NH0(RCELL,RHDIST,IHDIST)
C
C     COMPUTE ANGLE CHIP BETWEEN E VECTOR AND PERPENDICULAR TO JET
C             ANGLE MU   BETWEEN PROJECTED FIELD BPERP AND PERP TO JET
C
      IF(BPERP.EQ.0.) BPERP=0.001
      CMU=BY/BPERP
      IF (CMU.EQ.0.) GO TO 25
      SMU=(BZ*COSI-BX*SINI)/BPERP
      IF (SMU.EQ.0.) GO TO 30
      CHIP=ATAN2(-CMU,SMU)
      GO TO 40
   25 CHIP = 0.0
      GO TO 40
   30 CHIP = PI/2.
C      
C     COMPUTE THE FARADAY DEPTH OF THE CELL
C
   40 FARAD=ROT*NC0(RCELL,RCDIST,ICDIST)*BPARL*W/COSI
      IF(DEBUG.EQ.(.TRUE.)) WRITE(6,6) EMISS,CHIP,FARAD



    6 FORMAT(' TO LINE: EMISS=',E10.3,'    CHIP=',F6.2,'   FARAD=',
     *E10.3)
      END
C
C                        ************************
C                        *                      *
C                        *         B 3 D        *
C                        *                      *
C                        ************************
C
C               COMPUTE THREE-DIMENSIONAL MAGNETIC FIELD
C
      SUBROUTINE B3D(L,D,RCELL,BTOT,BPERP,BPARL,BX,BY,BZ)
C
C
      REAL L
      INTEGER*4 ISEED
      COMMON /AREA1/GAMMA,R,RHDIST,IHDIST,IPITCH,ILIST
      COMMON /AREA2/SINI,COSI,PSIR,SPSIR,CPSIR,ROT,BRAND,IFIELD
      COMMON /AREA5/BSHEAR,RSHEAR,IBGRD,PBGRD,CHIBGRD
      COMMON /DEBUG/DEBUG
      ISEED=31111
      IF(DEBUG.EQ.(.TRUE.)) WRITE(6,5) L,D,RCELL,R,CPSIR,SPSIR,COSI,SINI
    5 FORMAT(' NOW CALCULATING B COMPONENTS IN CELL L=',F6.2,
     *'    D=',F6.2,'    RCELL=',F6.2,'    R=',F6.2,/,
     *' WITH COS(PSIR)=',E10.3,' AND SIN(PSIR)=',E10.3,
     *' AND COS(I)=',E10.3,', SIN(I)=',E10.3)
C
C     BFACT WILL MAKE AVERAGE RANDOM FIELD VECTOR = BRAND*BNORM
C
      BFACT=4./SQRT(3.)
C
C     COMPUTE B IN THIS CELL FROM THE ADOPTED FORM:
C     IFIELD=1 => CHAN-HENRIKSEN FIELD (Ap.J. 241, 534 (1980))
C                 NEGLECTING THE SMALL RADIAL COMPONENT
C                 a particularly good assumption under shear - Oct 1995
C     IFIELD=2 => SIMPLE HELIX (CONSTANT PITCH)
C     IFIELD=3 => SHEARED HELIX (INVERSE CH)
C
      GO TO (1,2,3),IFIELD
      GO TO 4
    1 BPHI=CPSIR*(RCELL/R)
      GO TO 4
    2 BPHI=CPSIR
      GO TO 4
    3 BPHI=CPSIR*(R-RCELL)/R
      GO TO 4
C
C     Oct 1995 - REPLACE THE ABOVE BY SHEAR LAYER FIELD IF RCELL>=RSHEAR
C
    4 IF(RCELL.LT.RSHEAR) GO TO 59
      BZ=BSHEAR
      BPHI=0. 
      GO TO 60
C=======================================================================
C     CONVERT BZ,BPHI FIELD TO COMPONENTS BX,BY,BZ AND ADD RANDOM FIELDS
C     WHEN THE CELL IS NOT INSIDE THE SHEAR LAYER
C
C     USES THE V77 RAN FUNCTION



C     SEEMS MUCH FASTER ON SUNS THAN THE SUN rand FUNCTION!!
C
C
 59   BZ=SPSIR+BRAND*(RAN(ISEED)-0.5)*BFACT
 60   IF(RCELL.EQ.0) GO TO 10
      BX=BPHI*D/RCELL+BRAND*(RAN(ISEED)-0.5)*BFACT
      BY=BPHI*L/RCELL+BRAND*(RAN(ISEED)-0.5)*BFACT
      GO TO 20
C=======================================================================
C     SPECIAL CASE FOR CELL ON THE AXIS
C
   10 BX=0.
      BY=BPHI+BRAND*RAN(ISEED)
C=======================================================================
C
C     BTOT IS NORMALISED MAGNETIC FIELD STRENGTH
C
   20 BTOT=SQRT(BX**2 + BY**2 + BZ**2)
C
C     BPERP IS NORMALISED MAGNETIC FIELD STRENGTH PERP TO SIGHTLINE
C
      BPERP=SQRT((BZ*COSI-BX*SINI)**2+BY**2)
C
COMPUTE BPARL = COMPONENT OF FIELD PARALLEL TO LINE OF SIGHT 
C
      BPARL=BX*COSI + BZ*SINI
C
C     BPERP IS NORMALISED MAGNETIC FIELD STRENGTH PERP TO SIGHTLINE
C

      IF(DEBUG.EQ.(.TRUE.)) WRITE(6,6) BPERP 
    6 FORMAT(' BPERP =',E10.3)
      IF(DEBUG.EQ.(.TRUE.)) WRITE(6,7) BTOT
    7 FORMAT(' INITIAL BTOT =',E10.3)
      IF (BPERP.GT.BTOT) BTOT=BPERP
      IF(DEBUG.EQ.(.TRUE.)) WRITE(6,8) BTOT,BPERP,BPARL
    8 FORMAT(' TO EFCELL:  BTOT=',E10.3,'  BPERP=',E10.3,'   BPARL=',
     *E10.3)
      RETURN
      END
C
C                        ************************
C                        *                      *
C                        *           G          *
C                        *                      *
C                        ************************
C
C                     COMPUTE PITCH ANGLE DISTRIBUTION 
C
C
      FUNCTION G(THETA,IPITCH)
C
C
      IF(IPITCH.NE.1) GO TO 400
      G=1.
      RETURN
  400 IF(IPITCH.NE.2) GO TO 500
      G=COS(THETA)
      RETURN



  500 CTHETA=COS(THETA)
      G=CTHETA**2
      RETURN
      END
C
C                        ************************
C                        *                      *
C                        *        N H 0         *
C                        *                      *
C                        ************************
C
C     COMPUTE NUMBER DISTRIBUTION OF HOT ELECTRONS WITH RADIUS IN JET
C
C
      FUNCTION NH0(RCELL,RHDIST,IHDIST)
C
C
      REAL NH0
   D  IF(IHDIST.NE.1) GO TO 100
      NH0=EXP(-(RCELL/RHDIST)**2)
      RETURN
  100 IF(IHDIST.NE.2) GO TO 200
      NH0=EXP(-RCELL/RHDIST)
      RETURN
  200 IF(RCELL.GE.RHDIST) GO TO 300
      NH0=0.
      RETURN
  300 NH0=1.
      RETURN
      END
C
C                        ************************
C                        *                      *
C                        *        N C 0         *
C                        *                      *
C                        ************************
C
C     COMPUTE NUMBER DISTRIBUTION OF COLD ELECTRONS WITH RADIUS IN JET
C
C
      FUNCTION NC0(RCELL,RCDIST,ICDIST)
C
C
      REAL NC0
      IF(ICDIST.NE.1) GO TO 100
      NC0=EXP(-(RCELL/RCDIST)**2)
      RETURN
  100 IF(ICDIST.NE.2) GO TO 200
      NC0=EXP(-RCELL/RCDIST)
      RETURN
  200 IF(RCELL.GE.RCDIST) GO TO 300
      NC0=0.
      RETURN
  300 NC0=1.
      RETURN
      END
C
C                        ************************
C                        *                      *



C                        *      L P L O T       *
C                        *                      *
C                        ************************
C
C               CONVERTS FLUX ARRAY DATA TO LINE-PRINTER PLOT
C
C
      SUBROUTINE LPLOT(NMAX,IMAX)
C
C
      REAL IMAX
      DIMENSION PLOT(51,132),P(10)
      COMMON /AREA3/PROF(5,136),PINT
      COMMON /AREA4/PI,ICDIST,RCDIST
      COMMON /AREA5/BSHEAR,RSHEAR,IBGRD,PBGRD,CHIBGRD
      COMMON/DEBUG/DEBUG
C
C     CHARACTERS FOR THE PRINTER PLOTS
C
      DATA BLANK,DASH,TIC,PLUS,STAR/' ','-','I','+','*'/
      DATA P(1),P(2),P(3),P(4),P(5),P(6),P(7),P(8),P(9),P(10)
     */'0','1','2','3','4','5','6','7','8','9'/
C
C     CLEAR THE PLOT IMAGE AND SET UP REFERENCE GRID
C
      DO 2 I=1,51
      DO 3 J=1,132
      PLOT(I,J)=BLANK
      IF(MOD((51-I),10).EQ.0) PLOT(I,J)=DASH
      IF(MOD((J-6),10).EQ.0) PLOT(I,J)=TIC
      IF(J.EQ.1.OR.J.EQ.132) PLOT(I,J)=TIC
      IF(MOD((51-I),10).EQ.0.AND.MOD((J-6),10).EQ.0) PLOT(I,J)=PLUS
    3 CONTINUE
    2 CONTINUE
C
C     COMPUTE PLOT POSITIONS FOR FLUX DENSITY POINTS
C
      DO 14 K=1,NMAX
      J=INT(PROF(1,K))+66
      I=51-INT((PROF(2,K)*50./IMAX)+0.5)
C
C     PUT CHARACTER * AT EACH TOTAL-INTENSITY PLOT POSITION
C
      PLOT(I,J)=STAR
C
C=======================================================================
C
C     OLDER CODE TO SHOW DEGREE OF POLARIZATION - changed 03-Oct-1995
C    
C      QNORM=PROF(3,K)/PROF(2,K)
C      UNORM=PROF(4,K)/PROF(2,K)
C      PNORM=SQRT(QNORM*QNORM+UNORM*UNORM)
C      I=51-INT((ABS(PNORM)*50.)+0.5)
C
C
C     REPLACED BY FOLLOWING CODE TO SHOW POLARIZED INTENSITY DIRECTLY
C
      QNORM=PROF(3,K)
      UNORM=PROF(4,K)



      PNORM=SQRT(QNORM*QNORM+UNORM*UNORM)
      I=51-INT((PNORM*50./IMAX)+0.5)
C
C=======================================================================
C
      PA=ATAN2(UNORM,QNORM)*180./(2.*PI)
      IPA=ABS(INT(PA/10.))+1
      PLOT(I,J)=P(IPA)
      IF (PA.GE.0.) GO TO 14
      IF (PLOT(I,J-2).NE.DASH) PLOT (I,J-1)=DASH
   14 CONTINUE
C
C=======================================================================
C
C     CONVERT PLOT IMAGE TO LPT OUTPUT
C
      WRITE(3,10) (PLOT(1,J), J=1,132)
   10 FORMAT(1H , 132A1)
      DO 15 I=2,51
      WRITE(3,11) (PLOT(I,J), J=1,132)
   11 FORMAT(1H ,132A1)
   15 CONTINUE
      RETURN
      END
C
C                        ************************
C                        *                      *
C                        *      P R I N T       *
C                        *                      *
C                        ************************
C
C                  PRINT OUT THE PARAMETERS OF THIS MODEL
C
C
      SUBROUTINE PRINT(JETID,PSIRD,GAMMA,R,IHDIST,RHDIST,IPITCH,
     *ROT,BRAND,IFIELD,ICDIST,RCDIST,DAT,RSHEAR,BSHEAR,
     *IBGRD,PBGRD,CHIBGRD,IMAX)
C
C
      CHARACTER*24 DAT
      CALL FDATE(DAT)
      WRITE(3,20) DAT
   20 FORMAT(' CELL SUMMATION JET EMISSION MODEL WITH FARADAY ROTATION',
     *'      AHB code epoch 04-Oct-95',6X,'executed on ',A24,/,/)
      GO TO (21,23,25) IFIELD
   21 WRITE(3,22)
   22 FORMAT(' CHAN-HENRIKSEN FIELD',/)
      GO TO 29
   23 WRITE(3,24)
   24 FORMAT(' SIMPLE HELICAL FIELD',/)
      GO TO 29
   25 WRITE(3,26)
   26 FORMAT(' SHEARED HELICAL FIELD',/)
   29 WRITE(3,30) PSIRD,BRAND,R,RSHEAR,JETID,BSHEAR,GAMMA
   30 FORMAT(' REFERENCE PITCH OF FIELD (DEG)     :',F8.2,10X,
     *       ' MAGNITUDE OF RANDOM FIELD CPT      :',F8.2,/,
     *       ' JET BOUNDARY RADIUS                :',F8.2,10X,
     *       ' SHEAR INNER RADIUS                 :',F8.2,/,
     *       ' INCLINATION OF JET TO SKY (DEG)    :',F8.2,10X,



     *       ' SHEAR FIELD STRENGTH               :',F8.2,/,
     *       ' HOT ELECTRON ENERGY INDEX GAMMA    :',F8.2)
      GO TO (31,33,35) IHDIST
   31 WRITE(3,32) RHDIST
   32 FORMAT(' HOT ELECTRONS GAUSSIAN WITH 1/ESQ WIDTH ', F6.2)
      GO TO 39
   33 WRITE(3,34) RHDIST
   34 FORMAT(' HOT ELECTRONS EXPONENTIAL WITH SCALE RADIUS ', F6.2)
      GO TO 39
   35 WRITE(3,36) RHDIST
   36 FORMAT(' HOT ELECTRONS UNIFORM IN TUBE OF INNER RADIUS ', F6.2)
   39 GO TO (40,42,44) IPITCH
   40 WRITE(3,41)
   41 FORMAT(' HOT ELECTRON PITCH ANGLE DISTRIBUTION UNIFORM')
      GO TO 50
   42 WRITE(3,43)
   43 FORMAT(' HOT ELECTRON PITCH ANGLE DISTRIBUTION COSINE ')
      GO TO 50
   44 WRITE(3,45) 
   45 FORMAT(' HOT ELECTRON PITCH ANGLE DISTRIBUTION COSINE SQUARED')
   50 WRITE(3,46) ROT,IMAX
   46 FORMAT(' ROTATION MEASURE PARAMETER         :',F8.2,10X,
     *       ' MAXIMUM INTENSITY                  :',F8.2)
      GO TO (61,63,65) ICDIST
   61 WRITE(3,62) RCDIST
   62 FORMAT(' COLD ELECTRONS GAUSSIAN WITH 1/ESQ WIDTH ', F6.2,/)
      GO TO 69
   63 WRITE(3,64) RCDIST
   64 FORMAT(' COLD ELECTRONS EXPONENTIAL WITH SCALE RADIUS ', F6.2,/)
      GO TO 69
   65 WRITE(3,66) RCDIST
   66 FORMAT(' COLD ELECTRONS UNIFORM IN TUBE OF INNER RADIUS ', F6.2,/)
   69 WRITE(3,70) IBGRD,PBGRD,CHIBGRD
   70 FORMAT(' BACKGROUND PARAMETERS              : INT: ',F8.2,
     *' POL FRACTION: ',F6.2,' P.A.: ',F6.2,/,/,/,/,/,/)
      RETURN
      END
C
C                        ************************
C                        *                      *
C                        *      R E V I S E     *
C                        *                      *
C                        ************************
C
C
C              CONTROLS REVISION OF INPUT PARAMETERS
C
C
      SUBROUTINE REVISE(PSIRD,JETID,NL,IEX)
C
C
      COMMON /AREA1/GAMMA,R,RHDIST,IHDIST,IPITCH,ILIST
      COMMON /AREA2/SINI,COSI,PSIR,SPSIR,CPSIR,ROT,BRAND,IFIELD
      COMMON /AREA4/PI,ICDIST,RCDIST
      COMMON /AREA5/BSHEAR,RSHEAR,IBGRD,PBGRD,CHIBGRD
   75 CALL INPUTS(PSIRD,JETID,NL,IEX)
      IF (IEX.NE.1) GO TO 80
      WRITE(6,201)
  201 FORMAT(' NUMBER OF PARAMETERS TO REVISE ?')



      READ(5,20) NREV
   20 FORMAT(I2)
   21 FORMAT(F6.2)
      DO 202 J=1,NREV
      WRITE(6,203)
  203 FORMAT(' REVISE PARAMETER NUMBER ? ')
      READ(5,20) IREV
      WRITE(6,204) IREV
  204 FORMAT(' NEW VALUE OF PARM #',I2, ' ?')
      GO TO (300,301,302,303,304,305,306,307,308,309,310,311,312,
     *313,314,315,316,317,318),IREV+1
      GO TO 202
 300  READ(5,20) IFIELD
      GO TO 202
 301  READ(5,21) PSIRD
      GO TO 202
 302  READ(5,21) JETID
      GO TO 202
 303  READ(5,21) R
      GO TO 202
 304  READ(5,21) NL
      GO TO 202
 305  READ(5,21) GAMMA
      GO TO 202
 306  READ(5,20) IHDIST
      GO TO 202
 307  READ(5,21) RHDIST
      GO TO 202
 308  READ(5,20) IPITCH
      GO TO 202
 309  READ(5,21) ROT
      GO TO 202
 310  READ(5,21) BRAND
      GO TO 202
 311  READ(5,20) ICDIST
      GO TO 202
 312  READ(5,21) RCDIST
      GO TO 202
 313  READ(5,21) BSHEAR
      GO TO 202
 314  READ(5,21) RSHEAR
      GO TO 202
 315  READ(5,21) IBGRD
      GO TO 202
 316  READ(5,21) PBGRD
      GO TO 202
 317  READ(5,21) CHIBGRD
      GO TO 202
 318  READ(5,20) ILIST
 202  CONTINUE
      GO TO 75
 80   RETURN
      END
C
C                        ************************
C                        *                      *
C                        *      I N P U T S     *
C                        *                      *
C                        ************************



C
C
C               PUTS PARAMETERS UP ON TERMINAL FOR REVIEW
C
      SUBROUTINE INPUTS(PSIRD,JETID,NL,IEX)
C
C
      COMMON /AREA1/GAMMA,R,RHDIST,IHDIST,IPITCH,ILIST
      COMMON /AREA2/SINI,COSI,PSIR,SPSIR,CPSIR,ROT,BRAND,IFIELD
      COMMON /AREA3/PROF(5,136),PINT
      COMMON /AREA4/PI,ICDIST,RCDIST
      COMMON /AREA5/BSHEAR,RSHEAR,IBGRD,PBGRD,CHIBGRD
      WRITE(6,1) IFIELD, PSIRD
      WRITE(6,2) JETID,R,NL
      WRITE(6,3) GAMMA,IHDIST,RHDIST
      WRITE(6,4) IPITCH,ROT,BRAND,ICDIST
   
   WRITE(6,5) RCDIST,BSHEAR,RSHEAR,IBGRD,PBGRD,CHIBGRD,ILIST
      WRITE(6,6)
    1 FORMAT(' ',/,
     *       ' PARMS:',/,
     *       '  0 TYPE OF FIELD  1=CH  2=HELIX  3=SH.HELIX  : ',I8,/,
     *       '  1 REFERENCE PITCH OF FIELD (DEG)            : ',F8.2)
    2 FORMAT('  2 INCLINATION OF JET TO SKY (DEG)           : ',F8.2,/,
     *       '  3 JET BOUNDARY RADIUS                       : ',F8.2,/,
     *       '  4 NUMBER OF SIGHTLINES                      : ',I8)
    3 FORMAT('  5 ELECTRON ENERGY INDEX GAMMA               : ',F8.2,/,
     *       '  6 HOT ELECTRON DIST - 1=GAUSS 2=EXP 3=TUBE  : ',I8,/,
     *       '  7 1/ESQ (GAUSS), SCALE (EXP), INNER (TUBE) R: ',F8.2)
    4 FORMAT('  8 PITCH ANGLE DIST - 1=UNIFORM 2=COS 3=COSQ : ',I8,/,
     *       '  9 ROTATION MEASURE PARAMETER                : ',F8.2,/,
     *       ' 10 RANDOM FIELD SCALE MAGNITUDE              : ',F8.2,/,
     *       ' 11 COLD ELECTRON DIST - 1=GAUSS 2=EXP 3=TUBE : ',I8)  
    5 FORMAT(' 12 1/ESQ (GAUSS), SCALE (EXP), INNER (TUBE)  : ',F8.2,/,
     *       ' 13 SHEAR LAYER FIELD MAGNITUDE               : ',F8.2,/,
     *       ' 14 SHEAR LAYER INNER RADIUS                  : ',F8.2,/,
     *       ' 15 BACKGROUND INTENSITY                      : ',F8.2,/,
     *       ' 16 BACKGROUND FRACTIONAL POLARIZATION        : ',F8.2,/,
     *       ' 17 BACKGROUND POLARIZATION ANGLE             : ',F8.2,/,
     *       ' 18 LINE FLUX LIST ? - 0=NO 1=YES 2=DEBUG     : ',I8,/)
    6 FORMAT(' OPTIONS ARE: 1=REVISE   2=EXECUTE   3=STOP',/)
      READ(5,7) IEX
    7 FORMAT(I2)
      RETURN
      END


